
*-----------------------------------------------
* SPICE Netlist Generated by TDK Corporation 
* TDK P/N: ACM2012-900-2P (Common Mode Filter) 
* Property: Zc(at 100MHz) = 90 [ohm] 
* Model Generated on Dec. 04, 2006 
*-----------------------------------------------
* External Node Assignments: 
* 
*  1 ---@@@--- 4 
*       === 
*  2 ---@@@--- 3 
* 
*-----------------------------------------------
.SUBCKT ACM2012_900_2P 1 2 3 4 
C11 11 12 0.84p 
C12 21 22 0.84p 
C13 31 32 0.84p 
C14 41 42 0.84p 
C21 12 42 0.02p 
C22 22 32 0.02p 
C41 12 22 0.95p 
C42 32 42 0.95p 
K11 L21 L22 0.999999999 
L11 11 12 3.1n 
L12 21 22 3.1n 
L13 31 32 3.1n 
L14 41 42 3.1n 
L21 12 42 200n 
L22 22 32 200n 
R11 11 12 14 
R12 21 22 14 
R13 31 32 14 
R14 41 42 14 
R21 12 42 880 
R22 22 32 880 
R31 1 11 0.065 
R32 2 21 0.065 
R33 3 31 0.065 
R34 4 41 0.065 
R41 12 22 1G 
R42 32 42 1G 
R51 1 4 1G 
R52 2 3 1G 
.ENDS ACM2012_900_2P 
*-----------------------------------------------

Fig. 1. Model of "ACM2012-900-2P" 

•  Abstract 

Given that circuit design by simulators has recently become 

popular, TDK has released TDK SPICE Netlist Library for 

designing electronic circuits by circuit simulators. The library 

includes a set of equivalent circuit models of TDK electronic 

components written in SPICE Netlist format. Using this library, 

parasitic factors existing in electronic components can be considered 

in the circuit simulation. 

This application note describes the TDK SPICE Netlist Library, 

using it in SPICE simulators, and its basic applications. 

• Library introduction 

If an electronic component is simulated as an ideal one in a high 

frequency circuit, the simulated result does not match the actual 

property because real components have some parasitic factors whose 

effects cannot be disregarded at high frequencies. For designing high 

frequency circuits, TDK has released TDK SPICE Netlist Library, 

which includes a set of equivalent circuit models of TDK electronic 

components. Some of the parasitic factors in electronic components 

are considered in the equivalent circuit. Using this library, the actual 

property of electronic components can be considered in circuit 

simulation. All the models in the library are written in the subcircuit 

format of a generic SPICE Netlist. Since many circuit simulators 

including SPICE can interpret this format, this library can be used in 

many different simulators. 

This library contains models of ferrite beads (145 products), 

three-terminal filters (70 products), common-mode filters (31 

products), and pulse transformers (6 products) as of February 2007. 

• How to use the library 

 

To obtain the library 

The library can be downloaded from the TDK website at 

http://www.tdk.co.jp/etst. Since the downloaded file is compressed 

into a zip format, you will need to unzip the file before use. Read the 

readme file, which includes information about the library, before 

you start using the library. 

Model library 

Using a model from the library as an example, basic library use is 



2 

explained below. A model file is prepared for each product. 

Although the file extension is .mod, the file contents can be viewed 

and edited with a text editor because the file is in ASCII format. 

Here, a common-mode filter for differential signal lines 

"ACM2012-900-2P" is used as an example. A model of this product 

is described in the file, "ACM2012-900-2P.mod," and file contents 

are shown in Fig. 1. This model is written in a subcircuit format of a 

generic SPICE Netlist. Subcircuit is used to define the function of a 

specific circuit as a circuit package. A defined subcircuit can be 

applied whenever and wherever needed. Subcircuit is useful when a 

specific circuit is put in two or more places, or when part of the 

whole circuit is replaced by various subcircuits, to examine its 

properties. 

Figure 1 shows the model details. Lines that start with an asterisk 

(*) are comment lines and not used in simulation. These lines 

describe the product name, product category, typical properties of 

the product, external node assignment of subcircuit, and other 

information. The subcircuit definition is from the line of 

".SUBCKT" to ".ENDS". Between these lines, the equivalent circuit 

of ACM2012-900-2P is written in Netlist format. The actual 

electronic property can be considered in circuit simulation because 

some parasitic factors in electronic components are considered in the 

equivalent circuit. Information about the equivalent circuit is 

omitted in this application note. 

External node assignment 

In Fig. 1, the following expression defines the subcircuit: 

.SUBCKT ACM2012_900_2P 1 2 3 4 

The expression indicates that the subcircuit name is 

"ACM2012_900_2P" and 1, 2, 3, and 4 are assigned as the external  

nodes. Since the subcircuit is connected to the external circuit 

through these nodes, correctly recognizing each node 

and connecting them to the external circuit is important. 

In this example, external nodes 1, 2, 3, and 4 correspond 

to terminals as specified by the following diagram: 

*  1 ---@@@--- 4 
*       === 
*  2 ---@@@--- 3 

The diagram shows that each pair of nodes, such as 

nodes 1 and 2 or nodes 4 and 3, need to be connected to 

I/O nodes of the differential signal to make this circuit a 

common-mode filter. Note that the external node assignment is 

different in each product. The diagram in the model or the readme 

file included in the library provides information. 

Using the library 

This section describes how to use the library in a SPICE simulator. 

A case in which Netlist is used to input a circuit to a simulator is 

explained here. For information about SPICE Netlist, consult 

http://bwrc.eecs.berkeley.edu/Classes/IcBook/SPICE or the manual 

for the simulator being used. 

Two operations are necessary to use the model. The first operation 

is to copy the contents of the model file and paste it in the Netlist of a 

circuit where the model will be used. Some simulators can include 

an external file into a Netlist. In such simulators, a model file can be 

included into a Netlist without copying and pasting. The second is to 

add a description of calling the subcircuit in the Netlist. The second 

operation is explained in detail as follows: 

In SPICE Netlist, a device whose name starts with X is used for a 

subcircuit call. For example, the following description is used to call 

a subcircuit named ACM2012_900_2P: 

XCMF 12 22 23 13 ACM2012_900_2P 

Although the device name is XCMF in the example, any other 

names starting with X can be used. Four numbers following XCMF 

show node names of the external circuit to which the subcircuit is 

connected. In the above example, external nodes 1, 2, 3, and 4 of the 

subcircuit are connected, respectively, to nodes 12, 22, 23, 13 of the 

external circuit. Note that the name of the external node in a 

subcircuit may differ from the name of the node of the external 

circuit to which the subcircuit is connected. The important issue is 

not the node name itself, but the order of description. In this example, 

Fig. 2. Connections between the subcircuit and external  circuit 

***<< simulation of USB2.0 circuit >>************
*---< main circuit >---* 
V1 11 0 PULSE 0.0 0.8 0.0N 0.4N 0.4N 1.68N 4.16N
V2 21 0 PULSE 0.8 0.0 0.1N 0.4N 0.4N 1.68N 4.16N
RIN1 11 12 45 
RIN2 21 22 45 
XCMF 12 22 23 13 ACM2012_900_2P 
T1 13 0 14 0 Z0=45 TD=2N 
T2 23 0 24 0 Z0=45 TD=2N 
ROUT1 14 0 45 
ROUT2 24 0 45 
.TRAN 0.05N 10N 
.PRINT TRAN V(14) V(24) 
* 
*---< subckt >---* 
*-----------------------------------------------
* SPICE Netlist Generated by TDK Corporation 
* TDK P/N: ACM2012-900-2P (Common Mode Filter) 
* Property: Zc(at 100MHz) = 90 [ohm] 
* Model Generated on Dec. 04, 2006 
*-----------------------------------------------
* External Node Assignments: 
* 
*  1 ---@@@--- 4 
*       === 
*  2 ---@@@--- 3 
* 
*-----------------------------------------------
.SUBCKT ACM2012_900_2P 1 2 3 4 
C11 11 12 0.84p 
C12 21 22 0.84p 
C13 31 32 0.84p 
C14 41 42 0.84p 
C21 12 42 0.02p 
C22 22 32 0.02p 
C41 12 22 0.95p 
C42 32 42 0.95p 
K11 L21 L22 0.999999999 
L11 11 12 3.1n 
L12 21 22 3.1n 
L13 31 32 3.1n 
L14 41 42 3.1n 
L21 12 42 200n 
L22 22 32 200n 
R11 11 12 14 
R12 21 22 14 
R13 31 32 14 
R14 41 42 14 
R21 12 42 880 
R22 22 32 880 
R31 1 11 0.065 
R32 2 21 0.065 
R33 3 31 0.065 
R34 4 41 0.065 
R41 12 22 1G 
R42 32 42 1G 
R51 1 4 1G 
R52 2 3 1G 
.ENDS ACM2012_900_2P 
*-----------------------------------------------
* 
.END 

Fig. 5. The Netlist to simulate the circuit shown in Fig. 3

the connecting relation between the internal construction of the 

subcircuit and the external circuit is shown in Fig. 2. A pair of nodes, 

12 and 22, and another pair of nodes, 13 and 23, of the external 

circuit work, respectively, as I/O nodes of the differential signal. 

Recognizing the external node assignment of the subcircuit referring 

to the diagram and correctly connecting them to the external circuit 

is important. 

• Basic library applications 

Basic library applications are explained below. Let us consider a 

differential transmission circuit of USB 2.0 as shown in Fig. 3. The 

USB transmission line should be a pair of coupled lines, but this 

example uses two single-end lines excited in negative phase to 

model a differential circuit for ease of understanding. V1 and V2 are 

voltage pulse sources, RIN1 and RIN2 are output impedance of the 

sources, T1 and T2 are loss-less transmission lines, ROUT1 and 

ROUT2 are load resistances, and XCMF is a common-mode filter. 

The italic number in the figure indicates the node name. 

V1 and V2 are excited in the anti-phase and compose a differential 

signal source with RIN1 and RIN2. The actual signal source has 

Fig. 3. Differential transmission circuit 

Fig. 4. The output signal pattern from signal source 

non-zero impedance. But the output impedance of a voltage source 

is zero in simulation. Hence, the output impedance of the source is 

represented by RIN1 and RIN2. If there is time lag between 

anti-phase pulse signals excited by V1 and V2, skew (time lag 

between differential signals) is generated and a common-mode 

(in-phase) noise is added to the differential signal. Since a 

common-mode noise causes radiation noise, this effect should be 

suppressed as much as possible by using a common-mode filter or 

other measures. When a filter not suitable for the system is applied, 

the signal pattern will be distorted and transmission errors will occur. 

Here, the circuit is simulated by transient analysis function of SPICE, 

showing a common-mode filter suppresses a common-mode noise 

generated by skew and an unsuitable common-mode filter distorts 

the signal pattern. 

In high-speed USB 2.0, data is transmitted at 480 Mbps using a 

differential signal with voltage amplitude of 0.4 Vp-p, resulting in a 

time span of about 2.08 ns per bit. Supposing the rise and fall time of 

the pulse is 0.4 ns, the time lag of 0.1 ns is set in V2 to generate skew 

in order to add a common-mode noise. The output voltage pattern 

from the signal source (pulse source + output impedance) is shown 

in Fig. 4. The conditions of V1 and V2 are set to make the signal 

patterns of node 12 and 22 as in this figure. 

USB 2.0 specifications have a differential characteristic 

impedance of 90 ohm. The characteristic impedance of each 

single-end line is set to 45 ohm to model this. The output impedance 

of the signal source, characteristic impedance of the transmission 

line, and load resistance are also 45 ohm. 

Figure 5 shows the Netlist to simulate the circuit shown in Fig. 3. 

The main circuit and analysis commands are written in the top-half 

and the model of the common-mode filter is written in the 

bottom-half of the Netlist. Signal sources, transmission lines, and 

resistors are generic SPICE elements, while the common-mode filter 

is a model of "ACM2012-900-2P" included in the library. Since the 

input nodes of the differential signal to the common-mode filter are 

nodes 12 and 22 and output nodes of the signal from the filter are 

nodes 13 and 23, as shown in Fig. 3, the subcircuit call is described 

as "XCMF 12 22 23 13 ACM2012_900_2P". The electric length 

(delay time) of the transmission line is supposed to be 2 ns. The 

analysis commands are specified to perform a transient analysis to 

10 ns with a time step of 0.05 ns and to output voltage signal patterns 

at nodes 14 and 24. 

As shown in Fig. 5, the Netlist specifies that the voltage amplitude 

of pulse generated by V1 and V2 is 0.8 Vp-p, making the output 

signal pattern from the voltage source (pulse source + output 

impedance) determined only by load resistance, as shown in Fig. 4. 

This results from considering that the output voltage of the pulse 

source is divided by output impedance and load resistance. 

(c) without common-mode filter 

Fig. 6. The simulated results of transmitted signal pattern

(a) with common-mode filter "AMC2012-900-2P" 

(b) with other common-mode filter 

Fig. 7. Common-mode noise patterns 

Fig. 6 shows the simulated results of the Netlist in Fig. 5 by a 

SPICE simulator. For comparison, three results are shown: (a) with 

common-mode filter AMC2012-900-2P (common-mode 

impedance: 90 ohm @ 100 MHz; suitable for USB); (b) with other 

common-mode filter (common-mode impedance: 600 ohm @ 100 

MHz; not suitable for USB); (c) without common-mode filter. 

In all results, an output signal pattern responding to the input 

pulse is obtained after a time lag corresponding to the delay time (2 

ns) of the transmission line. Without a common-mode filter (Fig. 6c), 

skew generated at the signal source appears directly at the receiving 

point. In contrast, common-mode filters (Figs. 6a, b) suppress the 

skew very well. Using ACM2012-900-2P optimized for USB, a 

good transmitted signal pattern is obtained with little distortion of 

the signal pattern. Using another common-mode filter distorts the 

transmitted signal pattern and may cause transmission error. 

Selecting a suitable product for the system is important. 

Common-mode noise patterns are shown in Fig. 7 where the 

common-mode noise is calculated by a formula of (V(14)+V(24))/2. 

Figure 7 shows the result without common-mode filter as well as 

with ACM2012-900-2P. The amplitude of the common-mode noise 

is decreased from 100 mVp-p to 14 mVp-p using ACM2012-900-2P, 

which is expected to suppress radiation noise. 

• Conclusion 

This application note has described the TDK SPICE Netlist 

Library, using it in SPICE simulators, and its basic applications, 

explaining the basic issues of the library. The library can be used 

more conveniently, depending on the simulator used. Loading a 

model file using the file including function or making a circuit on a 

GUI using a circuit editor are two ways to benefit from such added 

convenience. Using a model of an actual IC as a driver or receiver 

can enable circuit simulation matching the actual properties. A 

model of a transmission line considering loss and coupling is also 

effective. Your simulator manual provides details. 

Please visit the website of our technical support tools at 

http://www.tdk.co.jp/etst.

Cautionary notes 

� The data in this application note and TDK SPICE Netlist Library do not guarantee product features. 

� TDK is not responsible for losses of any kind arising from using the contents of this application note or TDK SPICE Netlist Library.

� The contents of this application note are as of February 2007. 

� The contents of this application note and TDK SPICE Netlist Library are subject to change without notice. 


